

Un amplificateur en classe A, avec transformateur de sortie

J'invite le lecteur à consulter le site pour des informations complémentaires.

Page d'accueil du site Internet :

[page d'accueil](#)

d'autres pdf, sur différents sujets :

[liste des PDF](#)

L'amplificateur en classe A est connu pour sa simplicité, mais aussi pour son rendement médiocre. On peut améliorer ce dernier en plaçant un transformateur en liaison avec la charge. La théorie montre que le rendement atteint alors 50 %, dans des conditions de niveau de signal maximum. Nous allons vérifier cela par la simulation.

1) Etude de l'amplificateur en classe A, sortie sur transformateur

Usuellement appliqué sur l'émetteur commun, la classe A signifie un courant de polarisation qui fait travailler le transistor au milieu de ses caractéristiques $I_C = f(V_{CE})$ à I_B constant. Ici, la charge R_L est connectée via un transformateur, dont le primaire est parcouru par le courant collecteur.

Nous allons considérer ce schéma d'étude, simplifié au maximum.

Tension d'alimentation : E.


C_E , coupe la composante continue, et laisse passer le signal.

R_L représente la charge ; R_B assure la polarisation du transistor.

Le transformateur est parfait.

Rappelons les propriétés du transformateur :

$$m = n_2/n_1 \quad \vec{V}_2 = m\vec{V}_1 \quad \text{Impédance ramenée au primaire : } R_L/m^2$$




1.a) Polarisation

Dans un but de simplification et de linéarisation, la diode de la jonction B-E du transistor est modélisée par la caractéristique ci-contre :

V_S représente la tension de seuil (appelée parfois tension de déchet).

Pour une polarisation à $V_{BE} > V_S$, la jonction est passante et présente une caractéristique courant tension linéaire. On pose $r : \Delta V_{BE}/\Delta I_B$.



Le point de polarisation (I_{B0} , V_{BE0}) est donné par l'intersection entre cette caractéristique


d'entrée et la droite d'attaque $I_B = \frac{E - V_{BE}}{R_B}$.

On peut donc régler la polarisation par R_B .

On a, en statique : $I_C = \beta I_B$.

La valeur de I_{B0} donne I_{C0} .

Le primaire impose la droite de charge statique : $V_{CE} = E$.



1.b) Amplification en petits signaux

En dynamique, l'amplification en tension de l'émetteur commun, est $v_C/v_{in} = -g_m R_{EQ}$, (en idéalisant le transistor par une source de courant $i_C = g_m v_{BE}$) avec R_{EQ} la résistance équivalente vue par le collecteur, soit ici R_L/m^2 . On déduit $v_C/v_{in} = -g_m R_L/m^2$, et par suite $v_{out}/v_{in} = -g_m R_L/m$.

Remarquons, pour l'utilisateur, que le signe - est arbitraire car il suffit de permuter le sens d'un des enroulements pour inverser le signe.

L'équation de la droite de charge dynamique est : $i_C = -\frac{V_{CE}}{R_{EQ}}$, soit $= -v_{CE} \frac{m^2}{R_L}$, en ayant pour origine le

point de polarisation.

Le graphe illustre clairement l'amplification : une petite variation de tension autour de V_{BE0} entraîne une grande variation au potentiel de collecteur autour de V_{CE0} , et par suite un signal transmis en V_{out} .

1.c) Rendement

La variation maximale au collecteur est $\pm E$ autour de $V_{CE0} = E$.

Par simple lecture du graphe on a : $I_{C0} = m^2 (E/R_L)$.

En régime sinusoïdal, on a : $v_{CE}(t) = V_{CEmax} \sin \omega t$ et $i_C(t) = I_{Cmax} \sin \omega t$.

La puissance dans $R_L =$ puissance communiquée par le transformateur $= \frac{V_{CEmax} I_{Cmax}}{2}$, soit, dans ces

conditions d'amplitude maximale : $\frac{E I_{C0}}{2}$. La puissance délivrée par l'alimentation est $E I_{C0}$.

Le rendement atteint alors 50 %.

2) Simulations de vérification

Dans un premier temps, on réalise une simulation simpliste pour vérifier cette étude.


2.a) modélisation du transistor parfait

On modélise le transistor par le montage ci-contre.

On a en statique $I_C = \beta I_B$ et en dynamique $i_C = \beta i_B$. Même β .

L'amplification en tension de l'émetteur commun, en petits signaux autour du point de fonctionnement,

s'écrit également $\frac{v_C}{v_{in}} = -\frac{\beta R_{EQ}}{r} = -\frac{\beta R_L}{m^2 r}$. Après le transformateur, on a : $\frac{v_{out}}{v_{in}} = -\frac{\beta R_L}{m r}$.




2.b) Application numérique

Le circuit simulé fait apparaître ce transistor simplifié dans le schéma d'origine. On donne :

$E = 30 \text{ V}$; $C_e = 100 \mu\text{F}$; $R_B = 13,3 \text{ k}\Omega$; $R_L = 8 \Omega$

Pour des raisons de comparaison avec le transistor 2N 3055 donné dans le paragraphe suivant, on a : $r = 14 \Omega$; $V_S = 0,68 \text{ V}$; $\beta = 189$.

Transformateur : $L_p = 90 \text{ mH}$; $L_s = 10 \text{ mH}$; $K = 1$.



Remarque : par la relation $m = \sqrt{\frac{L_s}{L_p}}$, on déduit $m = 1/3$. Le

transformateur est monté en abaisseur de tension (d'un coefficient 3).

La charge ramenée au primaire est $R_{EQ} = R_L/m^2 = 72 \Omega$.

L'amplification attendue est de $-\frac{189 \times 72}{14} = -972$ coté collecteur et

324 coté secondaire.

50 % de rendement impose $I_{C0} = m^2 (E/R_L) = (1/9) (30/8) = \mathbf{0,416 \text{ A}}$, et donc $I_{B0} = 2,2 \text{ mA}$. Ceci est assuré par la valeur numérique de R_B .

Pour avoir l'amplitude maximale de 30 V en C (soit 10 V en out), il faut, en entrée, une amplitude de $30/972 = 30,86 \text{ mV}$.

La valeur de $V_S = 0,68 \text{ V}$ permet d'avoir le courant de base dans l'intervalle $[0 ; 2 I_{B0}]$.

βI_B est assuré par une source de courant commandée par un courant.

Ampli en classe A avec transfo de sortie

* fichier classeAsimplifie.cir

* circuit :

Valim E 0 30 ; alimentation

Vpola Ep 0 30 ; alimentation

Ve in 0 AC=1 sin (0 30.86m 10k 300m)

Ce in B 100u

RB Ep B 13.3k

* transistor équivalent

r B Bi 14

.param VS = 0.68

VS Bi 0 DC={VS} ;

.param beta=189

Fcol C 0 VS {beta}

Lp E C 90mH

Ls out 0 10mH

K Lp Ls 1

RL out 0 8 ; charge

.OP

.AC DEC 100 1 100MEG

.TRAN 0.2u 300.2m 299.9m 0.2u

.probe

.end

Netlist, prête à simuler


2.c) Polarisation

L'analyse OP confirme les réglages : 30 V au collecteur.

NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE	NODE	VOLTAGE
(B)	.7108	(C)	30.0000	(E)	30.0000	(Bi)	.6800
(in)	0.0000	(out)	0.0000				

2.d) Réponse harmonique

L'amplification attendue, en moyenne fréquence, est $\frac{V_C}{V_{in}} = -972$.



Réponse harmonique $20 \log \left| \frac{V_C}{V_{in}} \right|$: 29,75 dB et en linéaire : 972.

En basse fréquence, 2 phénomènes interviennent :

- Ce et r forment un filtre passe haut de coupure $\frac{1}{2\pi r C_e} = \frac{1}{2\pi \cdot 14 \cdot 100\mu} = 113,68 \text{ Hz}$,


- la fréquence à partir de laquelle le transformateur assure le transfert d'impédance : $\frac{R_L}{2\pi L_s} = \frac{8}{2\pi \cdot 10\text{m}} = 127 \text{ Hz}$. Pour des fréquences inférieures, la charge ramenée n'est pas 72Ω , mais augmente. L'amplification est donc supérieure à 972.

Ces 2 phénomènes cumulés aboutissent à une fréquence de coupure haute à 187 Hz.

Measurement	Value
Cutoff_Highpass_3dB(V(C))	187.85887

2.e) Réponse transitoire, signal d'entrée sinusoïdal


On choisit une fréquence de 10 kHz (car dans la bande passante), et l'amplitude à 30,8 mV.



Potentiel de collecteur et courant de collecteur du transistor parfait.

Le délai placé sur la source V_{in} permet de visualiser la polarisation : $I_{C0} = 0,416 \text{ A}$, $V_{C0} = 30 \text{ V}$.
Puis quand le signal V_{in} est appliqué, le courant va de 0 A à 0,832 A tandis que le potentiel collecteur va de 60 V à 0 V respectivement.

Les mêmes courbes placées en XY montrent la droite de charge dynamique :



Droite de charge dynamique : centrée sur 30 V, 0,416 A, $\Delta V_C / \Delta I_C = 72 \Omega$

On peut faire le **bilan des puissances** :

On simule sur une plus longue durée, pour être en régime établi.

6,25 W sont dissipés par R_L (10 V d'amplitude, sur 8 Ω)

et 12,45 W délivrés par l'alim.

On a bien **50 %** de rendement. Le « transistor » dissipe 6,25 W.

r	Trace Name	Y1
	X Values	303.000m
	AVGX(W(RL),2m)	6.2455
	-AVGX(W(Valim),2m)	12.449

3) Simulations du montage

3.a) Modèle du 2N3055

Ce transistor de puissance, très ancien et très connu des électroniciens, a son modèle Pspice donné sous :

<https://www.onsemi.com/support/design-resources/models?rpn=2N3055>

Il est recopié dans la netliste :

Son format est TO3 :



Il peut supporter $V_{CE} = 60 \text{ V}$ en base ouverte, 70 V en base connectée avec une résistance :

MAXIMUM RATINGS

Rating	Symbol	Value	Unit
Collector-Emitter Voltage	V_{CEO}	60	Vdc
Collector-Emitter Voltage	V_{CER}	70	Vdc

C'est donc compatible avec nos simulations.

Ampli en classe A avec transfo de sortie

* fichier classeA_2n3055.cir

* circuit :

Valim E 0 30 ; puissance

Vpola Ep 0 30 ; polar

Ve in 0 AC=1 sin (0 50m 10k 300m)

Ce in B 100u

RB Ep B 13.3k

Q1 C B 0 Q2N3055

Lp E C 90mH

Ls out 0 10mH

K Lp Ls 1

RL out 0 8 ; charge

.OP

.AC DEC 100 1 1MEG

.TRAN 0.2u 315m 299.9m 0.2u

.MODEL Q2n3055 npn

+IS=2.37426e-14 BF=129.119 NF=0.85 VAF=31.1252

+IKF=0.990922 ISE=2.47498e-10 NE=1.89002 BR=1.01252

+NR=0.924456 VAR=254.624 IKR=2.70227 ISC=2.47498e-10

+NC=2.90624 RB=3.66609 IRB=0.1 RBM=0.1

+RE=0.000352673 RC=0.0764459 XTB=1.34801 XTI=1.07207

+EG=1.206 CJE=9.03089e-08 VJE=0.513954 MJE=0.59999

+TF=1e-8 XTF=1.36696 VTF=1.02605 ITF=0.987296

+CJC=5e-10 VJC=0.400243 MJC=0.410238 XCJC=0.803124

+FC=0.661216 CJS=0 VJS=0.75 MJS=0.5

+TR=1e-7 PTF=0 KF=0 AF=1

* Model generated on Jan 24, 2004

* Model format: PSpice

.probe

.end

Netlist, prête à simuler

3.b) Polarisation

On retrouve la même polarisation : **$I_{C0} = 416 \text{ mA}$, $I_{B0} = 2,2 \text{ mA}$** .

La résistance $r_{pi} = 10,4 \text{ } \Omega$ (calculée par Pspice) est en série avec $R_B = 3,66$ (donné dans le .model), ce qui donne un total de $14,06 \text{ } \Omega$. Cela explique la valeur $r = 14 \text{ } \Omega$ du modèle simplifié pour présenter la même résistance.


On remarque la faible valeur de $r_0 : 145 \text{ } \Omega$. La conséquence est une perte d'amplification, comme montré au paragraphe suivant.

On a $\beta_{AC} = 168$.

Notons également $C_{BE} = 52,5 \text{ } \mu\text{F}$ et $C_{BC} = 68,6 \text{ pF}$.

NAME	Q1
MODEL	Q2N3055
IB	2.21E-03
IC	4.16E-01
VBE	6.68E-01
VBC	-2.93E+01
VCE	3.00E+01
BETADC	1.89E+02
GM	1.61E+01
RPI	1.04E+01
RX	3.50E+00
RO	1.45E+02
CBE	5.25E-07
CBC	6.86E-11
CJS	0.00E+00
BETAAC	1.68E+02
CBX/CBX2	1.68E-11
FT/FT2	4.88E+06

3.c) Réponse harmonique



Réponse harmonique : L'amplification $V(C) / V(in)$ est 574,7 en milieu de bande. En $V(out)$, toute la courbe est multipliée par m , soit 191,5 en milieu de bande.

La présence de r_0 fait chuter R_{EQ} à $72 \text{ } \Omega // 145 = 48,11 \text{ } \Omega$.

L'amplification en milieu de bande devient $168 \times 48,11 / 14,06 = \mathbf{574,8}$, ce qui est confirmé par la simulation AC.

Measurement	Value
Cutoff_Highpass_3dB(V(C))	155.15578
Cutoff_Lowpass_3dB(V(C))	116.53632k

Comme précédemment, la fréquence de coupure basse de 155 Hz est essentiellement créée par C_e .


En hautes fréquences, le transistor peut être remplacé par son modèle de Giacoletto.

La fréquence de coupure haute peut être estimée à :

$$\frac{1}{2\pi \text{ req } C_{eq}}$$

où $\text{req} = r_B // r_{\pi}$ et $C_{eq} \approx C_{BE} - A_v C_{BC}$

par application du théorème de Miller



Modèle de Giacoletto

L'application numérique donne :


$$\text{req} = 3,66 // 10,4 = 2,707 \text{ } \Omega \quad \text{et} \quad C_{eq} \approx 5,27 \cdot 10^{-7} + 574 \times 6,86 \cdot 10^{-11} = 5,66 \cdot 10^{-7} \text{ F}$$

soit une fréquence de coupure de 103,8 kHz, valeur proche de celle donnée par l'analyse AC.


3.d) Réponse temporelle, signal d'entrée sinusoïdal

Pour bénéficier de la dynamique maximale, il faut, théoriquement, disposer en entrée d'une amplitude égale à $30 \text{ V} / 574,7 = 52,2 \text{ mV}$. Sauf que ce calcul prend comme hypothèse que le système est linéaire, ce qui est loin d'être le cas.

Par exemple, à $V_{in} = 50 \text{ mV}$ les chronogrammes montrent une très nette distorsion :




Potentiel de collecteur et courant de collecteur du 2N3055. On a une très nette distorsion, et les valeurs min et max n'atteignent pas les maxima (repérés par des lignes vertes et bleues).



Droite de charge dynamique : $\Delta V_C / \Delta I_C = 72 \Omega$. On reconnaît nettement que la dynamique maximale n'est pas atteinte, malgré le fort niveau en entrée.

Pour illustrer que le système est non linéaire, traçons $V_{out} = f(V_{in})$ et $V(C) = f(V_{in})$:



Caractéristique de transfert : on est loin d'une droite de pente - 574,7 ou 191,5

La courbure montre clairement la non-linéarité (provoquée par le 2N3055 en forts signaux). Le dédoublement de la courbe illustre un léger déphasage à 10 kHz.

On peut faire le **bilan des puissances** :

On simule sur une plus longue durée, pour être en régime établi.

3,13 W sont dissipés par R_L et 9 W délivrés par l'alim.

On a **34,7 %** de rendement, loin des 50 % théorique.

Le transistor dissipe 5,88 W.

X Values	
AVGX(W(RL),2m)	3.1290
AVGX(W(Q1),2m)	5.8821
-AVGX(W(Valim),2m)	8.9915

Characteristic	Symbol	Max	Unit
Thermal Resistance, Junction-to-Case	$R_{\theta JC}$	1.52	$^{\circ}C/W$

Notons qu'en supposant une température ambiante de $25^{\circ}C$, et une température de jonction maximale de $200^{\circ}C$, la loi d'ohm thermique aboutit à $R < \Delta T / P = 175 / 5,88 \approx 30^{\circ}C/W$.

Sachant que $R_{THJ-C} = 1,52^{\circ}C/W$, il faut adjoindre un dissipateur de résistance thermique (associé au contact radiateur-boîtier) inférieure à $28,5^{\circ}C/W$.

Conclusion

Une étude théorique est souvent réalisée sur un schéma simplifié, afin d'établir, sans trop de difficulté, des équations et leurs résolutions.

Dans l'amplificateur classe A avec transformateur parfait, il est d'usage de retenir que le rendement maximum est de 50 % si on place un signal d'entrée suffisamment grand pour couvrir le maximum de dynamique en tension et en courant.

Ce document rappelle que cette valeur de 50 % n'est obtenue que si le système est linéaire, ce qui est une contradiction avec la notion de grands signaux dès qu'on travaille avec un transistor.

Dans notre simulation, on voit que même si on accepte une forte distorsion (valeur de 11% donnée par la commande .FOUR 10kHz V(RL)), le rendement est loin des 50 % : nous avons 34 %.

La charge $R_L = 8 \Omega$ laisse penser que ce schéma est un amplificateur « audio », la résistance R_L étant un modèle simplifié d'un haut-parleur. Si c'était vraiment le cas, il faudrait - a minima - compléter l'analyse par un modèle réaliste du transformateur. Or, le transformateur utilisé dans ces pages est purement théorique et parfait : on n'a pas tenu compte des résistances des enroulements, ni des fuites magnétiques, ni du comportement du circuit magnétique (hystérésis, saturation). Tous ces paramètres influent énormément sur la réponse harmonique, limitent la bande passante, et, par l'ajout de pertes, baissent le rendement : on s'éloigne encore des 50 %...

articles 1 à 43 : sur le livre

Tableau récapitulatif des articles PDF disponibles sur ce site

n°	titre	lien présentation	lien direct article
	Guide d'installation et d'emploi simplifié	présentation	document PDF
44	Exemples basiques et des exercices...	présentation	document PDF
45	Un exemple de circuit passif	présentation	document PDF
46	Un oscillateur Colpitts	présentation	document PDF
47	Compensation en fréquence des amplificateurs opérationnels	présentation	document PDF
48	Un amplificateur à transistors bipolaires	présentation	document PDF
49	Une bascule D Flip Flop CMOS	présentation	document PDF
50	Une porte XOR à transistors MOS	présentation	document PDF
51	Un VCO à 12 transistors MOS	présentation	document PDF
52	Une PLL à moins de 20 transistors MOS	présentation	document PDF
53	Un oscillateur à résistance négative	présentation	document PDF
54	Une charge électronique	présentation	document PDF
55	Un amplificateur en classe C	présentation	document PDF
56	Le monostable 74 123	présentation	document PDF
57	Un amplificateur en classe D	présentation	document PDF
58	Le transformateur en linéaire	présentation	document PDF
59	La loi d'ohm thermique	présentation	document PDF
60	Le transformateur en non linéaire	présentation	document PDF
61	Robustesse d'un oscillateur en anneau	présentation	document PDF
62	Une alimentation stabilisée	présentation	document PDF
63	Modélisation d'un haut-parleur	présentation	document PDF
64	Un synthétiseur de fréquence	présentation	document PDF
65	Un ampli audio de Sparkfun	présentation	document PDF
66	Simulation logique et analogique	présentation	document PDF
67	Un oscillateur à relaxation	présentation	document PDF
68	Lecteur de TAG RFID 125 kHz	présentation	document PDF
69	Diagramme de l'œil avec Pspice	présentation	document PDF
70	Un amplificateur hautes fréquences	présentation	document PDF
71	Une bizarrerie enfin expliquée...	présentation	document PDF
72	Comprendre le paramétrage de la FFT	présentation	document PDF
73	La relation de Bennett	présentation	document PDF
74	Simuler un circuit à plus de 20 transistors avec PSpice Eval	présentation	document PDF
75	Une horloge biphase sans recouvrement	présentation	document PDF
76	Quelques simulations sur la diode	présentation	document PDF
77	Un ampli classe A, avec transformateur de sortie	présentation	document PDF
78	Des stimuli pour PSpice	présentation	document PDF
79	Simuler le TL431 : zener ajustable	présentation	document PDF
80	Un ADC flash	présentation	document PDF
81	Une chaîne d'acquisition : S&H, ADC, DAC	présentation	document PDF
82	Un amplificateur 50 MHz	présentation	document PDF
83	Un dérivateur non inverseur	présentation	document PDF
84	Un amplificateur bipolaire avec push pull CMOS	présentation	document PDF
85	Rôle des répéteurs logiques dans un circuit intégré	présentation	document PDF
86	Un driver logique CMOS pour charge 50 ohms	présentation	document PDF
87	Des triggers de Schmitt et des applications	présentation	document PDF
88	Un filtre gaussien analogique	présentation	document PDF
89	Un générateur de bruit rose	présentation	document PDF

90	Un anémomètre à fil chaud : simulation comportementale	présentation	document PDF
91	Un oscillateur à pont de Wien stabilisé par CTN	présentation	document PDF
92	L'emballement thermique d'une diode	présentation	document PDF
93	Les puissances dans un amplificateur	présentation	document PDF
94	Asservissement de puissance dans une résistance	présentation	document PDF
95	Asservissement de la puissance émise par une antenne radio	présentation	document PDF
96	Un driver de LED de puissance	présentation	document PDF
97	Exploiter Pspice pour simuler des filtres numériques	présentation	document PDF
98	Un filtre en cosinus surélevé avec Pspice	présentation	document PDF
99	Effet de la température sur un amplificateur en classe A	présentation	document PDF
100	Un amplificateur à transistors JFET et bipolaires	présentation	document PDF
Supplément, hors article :			
mon cours « Electronique pour les communications numériques », polycopié couleur 201 pages en pdf			

[retour à l'écran d'accueil de ce site](#)